

# DIALTONE

**laten telefoneren met je MSX**

**MSX Club Magazine 35**

*Scanned, ocr'ed and converted by HansO, 2001*

Een leuke toepassing voor uw computer. Met DIALTONE zijn wij in staat om hem voor ons te laten bellen. Dat wil zeggen het nummer kiezen en draaien.

## **Liever niet solderen**

Ik las in een blad een artikel over bellen met de computer, maar omdat werkend te krijgen moest men eerst allerlei dingen aan de printerpoort hangen. Maar omdat sommige computers niet eens een printerpoort hebben, en de meeste computergebruikers, ik ook niet, liever niet aan hun computer solderen, heb ik dit programma geschreven.

## **Telefooncentrale**

Voorwaarde is wel dat je woonplaats een computergestuurde telefooncentrale heeft. Dit is te herkennen aan de bekende toontjes bij het bellen. Zo'n toon is opgebouwd uit twee sinusvormige frequenties, en die worden door de telefooncentrale omgezet in een nummer. Afhankelijk welk nummer je draait, wordt de ene of de andere toon hoger, en dat is precies wat dit programma doet. Je typt de naam of het nummer in, houdt de hoorn zo dicht mogelijk bij de luidspreker van de monitor of de tv.

## **Experimenteren**

Als het niet in een keer werkt dan moet je eens experimenteren met het volume e.d. Bij mij werkt het in ieder geval perfect. Achteraan het programma staan DATA regels. Daar kun je zelf namen en telefoonnummers in zetten. De volgende keer kun je dan gewoon de naam intypen en de computer belt dan diegene op. Er hoeft niet aan de telefoon of computer gerommeld te worden dus er kan ook niets beschadigd worden.

```

100 ' Bellen per computer
110 ' Door: David Simons
120 '     Bevrijding 107
130 '     7121 WT Aalten
140 '     Tel:05437-75986
150 '
160 SCREEN 0,,0:CLS:CLEAR 200:KEY OFF
170 FOR T = 0 TO 9:READ A1(T),A2(T):NEXT T
180 CLS:B$="":C$=""
190 A$="":PRINT "GEEF NAAM (RET IS NUMMER) :";
200 INPUT A$
210 IF A$ = "" THEN PRINT:PRINT "- IS PAUZE":INPUT "NUMMER:";C$:GOTO 270
220 RESTORE 670
230 READ B$,C$
240 IF B$= "*" THEN PRINT "NAAM NIET GEVONDEN...":PRINT:GOTO 190
250 IF A$= B$ THEN PRINT A$;" WORDT NU GEBELD...":PRINT:PRINT C$:GOTO
270
260 GOTO 230
270 FOR Z = 1 TO LEN(C$)
280 A$=MID$(C$,Z,1)
290 ' FOR X = 1 TO XXXX < PAUZE VOOR WACHTEN OP TOON
300 IF A$="-" THEN FOR X = 1 TO 1000:NEXT X,Z
310 IF A$="#" THEN A1=1209:A2=941:GOTO 370
320 IF A$="*" THEN A1=1447:A2=941:GOTO 370
330 A=VAL(A$)
340 ' FREQUENTIE BEREKENEN
350 ' x1 = TOON 1
360 ' x2 = TOON 2
370 R1=INT(111860!/A1(A))
380 R2=INT(111860!/A2(A))
390 M1=INT(R1/256)
400 M2=INT(R2/256)
410 L1=R1-256*M1
420 L2=R2-256*M2
430 ' HET EIGENLIJKE BELLEN
440 SOUND 0,L1
450 SOUND 1,M1
460 SOUND 2,L2
470 SOUND 3,M2
480 ' VOLUME INSTELLEN
490 SOUND 8,14
500 SOUND 9,10
510 ' LENGTE TOON
520 FOR T = 1 TO 80:NEXT T
530 SOUND 9,0:SOUND 8,0
540 ' PAUZE VOOR ANDERE TOON
550 FOR T = 1 TO 10:NEXT T
560 A$="":A=0:NEXT Z:GOTO 180
570 '
580 ' FREQUENTIE TONEN IN Hz
590 '
600 DATA 1336,941,1209,697,1336,697,1447,697
610 DATA 1209,770,1336,770,1447,770,1209,852
620 DATA 1336,852,1447,852
630 '
640 ' TELEFOONNUMMERS

```

```
650 ' NAAM,TELEFOONNUMMER
660 '
670 DATA PIETJE,070-3123456
680 DATA JANTJE,010-654321
690 DATA KLAASJE,020-123654
700 DATA HENK,050-456321
710 DATA *,*
720 ' *,* = EINDE ZOEKEN
```